Lagrangen kertojat ja ominaisarvot

Luentoesimerkki ti 19.3.02 HA

```
> with(LinearAlgebra):
> A:=<<17,6>|<6,8>>;
```

$$A := \begin{bmatrix} 17 & 6 \\ 6 & 8 \end{bmatrix} \quad (1)$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 20 \\ 5 \end{bmatrix}, \begin{bmatrix} 2 & -\frac{1}{2} \\ 1 & 1 \end{bmatrix} \quad (2)$$

```
>     f:=(x,y)->x^2+y^2:
> g:=(x,y)->17*x^2+12*x*y+8*y^2-100;
```

$$g := (x, y) \to 17 x^2 + 12 x y + 8 y^2 - 100 \quad (3)$$

```
> L:=(x,y,lambda)->f(x,y)+lambda*g(x,y);
```

$$L := (x, y, \lambda) \to f(x, y) + \lambda\, g(x, y) \quad (4)$$

```
> solve({diff(L(x,y,lambda),x)=0,diff(L(x,y,lambda),y)=0,g(x,y)=0},
  {x,y,lambda});
```

$$\left\{\lambda = -\frac{1}{20}, x = 2, y = 1\right\}, \left\{\lambda = -\frac{1}{5}, x = 2, y = -4\right\}, \left\{\lambda = -\frac{1}{5}, x = -2, y = 4\right\}, \left\{\lambda = -\frac{1}{20}, x = -2, y = -1\right\} \quad (5)$$

Tassa tapauksessa Lagrangen kerroin λ on ominaisarvojen kaanteisluvun vastaluku. (Vastaluku tulee vain siita, etta siirsimme
kaikki termit samalle puolelle.) Harjoitustehtavassa on tilaisuus testata hypoteesia uudestaan.

Katsotaan samantien numeerista ratkaisijaa. Se tarvitsee yleensa alkuarvot. Kokeillaan.

```
> yht:={diff(L(x,y,lambda),x)=0,diff(L(x,y,lambda),y)=0,g(x,y)=0};
```

$$yht := \{2 x + \lambda (34 x + 12 y) = 0, 2 y + \lambda (12 x + 16 y) = 0, 17 x^2 + 12 x y + 8 y^2 - 100 = 0\} \quad (6)$$

```
> fsolve(yht,{x=0,y=0,lambda=0});
```

$$\{\lambda = -0.05000000000, x = 2.000000000, y = 1.000000000\} \quad (7)$$

```
> fsolve(yht,{x=1,y=-1,lambda=-1});
```

$$\{\lambda = -0.2000000000, x = 2.000000000, y = -4.000000000\} \quad (8)$$